
ABSTRACT 
The quality and completeness of birth certificate data is essential to public health 
research, planning and services. State and Federal Maternal and Child health 
programs utilize birth certificate data to target mothers from particular 
racial/ethnic and age groups with low utilization of prenatal care or to investigate 
maternal characteristics associated with low birth weight, prematurity, infant 
mortality and other unfavorable birth outcomes.   
 
Every year, the CDC-National Center for Health Statistics (NCHS) sets up 
standards of acceptable proportions of invalid data fields reported as not 
classifiable (i.e., outside predetermined ranges), unknown or blank in birth 
certificate data. Since the vast majority of birth certificates are collected and 
submitted by hospitals, the Center for Health Statistics at the Arkansas 
Department of Health needed a report that can easily identify fields with high 
proportions of invalid data by hospital of birth. Such a report is important for 
administrators of vital records and field representatives who can contact these 
hospitals to provide educational seminars to nurses and clerks on appropriate 
methods in collecting and reporting birth data. 
 
This paper describes a SAS® program that can be used on any arbitrary set of 
birth data to provide a quick data quality tool. The project uses several features in 
the SAS programming language such as IF-THEN-ELSE statements, the 
SUMMARY procedure and ARRAY statements, but will mainly focus on the use 
of PROC FORMAT and PROC REPORT to color-code values of variables based 
on a decision rule1 (i.e., exceed NCHS acceptable standard of invalid data) using 
the Output Delivery System (ODS).  
 
INTRODUCTION 
Arkansas State laws require birth certificates to be completed for all births, and 
Federal law mandates national collection and publication of births and other vital 
statistics data2. The National Vital Statistics System (NVSS), the Federal 
compilation of this data, is the result of the cooperation between NCHS and State 
agencies to provide access to valuable statistical information from birth 
certificates. 
 
With the introduction of ODS, we thought of taking advantage of its strong 
reporting capabilities and the ability to create appealing tables, to craft a report 

                                                 
1 A similar approach is described by Karp (2003) 
2 National center for Health Statistics (NCHS)  

that can be used to identify hospitals that contribute to high quantities of invalid 
data.  
 
DESCRIPTION OF THE PROJECT    
Using birth data submitted by 54 states and territories, NCHS, every year, 
generates a “NATALITY NOT CLASSIFIABLE AND UNKNOWN DATA 
REPORT” and makes it available to Centers for Health Statistics and Vital 
Records or any other individual state agencies that collect, edit and publish vital 
statistics data.  
 
In 2003, NCHS selected 40 data fields to report on. The report includes the 
acceptable percentage of records (standard) that are not classifiable, blank and 
unknown. Standards are computed using criteria of multiplying the nationwide 
median for 1998 for a particular item, as a percentage of records received by 
NCHS, by a multiplier. The multiplier was 1.5 for 2003, which is 0.5 less than that 
of 2002. The following is an example of how the standard was computed for 
prenatal visits in different years. 
 
Prenatal 
Visits 

1998 Median 
(%) 

2000 STND = 
3X the 
Median for 
1998 

2001 STND = 
2.5X the 
Median for 
1998 

2002 STND = 
2X the 
Median for 
1998 

2003 STND = 
1.5X the 
Median for 
1998 

 1.99 5.97 4.98 3.98 2.99 
 
Since most births occur in hospitals and freestanding birth facilities, which are 
required to submit birth certificate data to the Arkansas Department of Health-
Division of Vital Records, we wanted to generate an Arkansas-specific report by 
hospital of birth using the same standards supplied by NCHS. We also wanted to 
identify fields where Arkansas exceeds the NCHS standard for that particular 
year and also identify hospitals that exceed that standard. One senior analyst 
suggested that the report highlights fields that may potentially exceed the 
standard for both the state and individual hospitals.  
 
For purposes of this paper, we have selected 8 birth certificate items that are 
looked at more often in analyzing preterm birth3. These are: clinical estimate of 
gestation, day last normal menses began, month last menses began, year last 
menses began, month prenatal care began, named month prenatal care began, 
prenatal visits and weight gain during pregnancy.  
 
BEFORE ODS 
Before we learned about ODS, we generated a similar report but used Microsoft 
Excel to present the tables.  We categorized the selected variables, summarized 
the data using PROC SUMMARY and exported the data table to Excel. In Excel, 
we customized some Visual Basic for Applications (VBA) scripts generated by 
running user-defined Excel macros to color-shade cells where values exceeded 
                                                 
3 Arkansas has been experiencing a decline in preterm birth rate in the last five years. The accuracy and completeness of 
pertinent data is essential to analyzing the significance of the trend.          

the standards. With the introduction of ODS and the powerful PROC REPORT, 
we thought of generating the report “by clicking only one button” entirely in a SAS 
environment.  
 
DESCRIPTION OF THE SAS PROGRAM         
 
Getting Started 
 
OPTIONS PAGENO = 1 CENTER PAPERSIZE = LEGAL ORIENTATION = LANDSCAPE;  
LIBNAME birth 'path-to-server/birth database'; 
%LET data = birth.birth2003; 
%LET path = C:\SCSUG;  
 

To be able to generate a report with all your summarized variables (numbers and 
percents) lined up in one page, use PAPERSIZE = LEGAL ORINETATION = 
LANDSCAPE options in the OPTIONS statement. The first %LET macro program 
statement, the simplest way to create and assign a value to a macro variable, is 
used to assign a value to the permanent birth data file and the second %LET is 
defining the path where you want to store your generated report (in a PDF file 
format, discussed later). 
 
DATA temp (keep = ostate bhosp_ bloc gest gest1 menda menda1 menmo 
                  menmo1 menyr menyr1 care care1 nammo nammo1  
                  visits visits1 wtgain_ wtgain1 rec_err); 
   set &data (pw = *****); 
   if bloc not in ('1', '2') then bhosp_ = 'Non-Hosp';  
 
This DATA step creates a birth work file and the KEEP =, a data step option, 
keeps the variables that you will be using throughout the program in the work 
dataset. The IF-THEN statement above recodes birth hospital numbers (bhosp_) 
for non-hospital births to ‘Non-Hosp’. This will give an idea on the quality of non-
hospital (mostly home births) birth certificate data.  
 
/********************************************** 
IF-THEN-ELSE Statements:  
CREATE NEW VARIABLES WHERE VALID DATA ENTRIES  
WILL BE ASSIGNED '0s', WHILE NOT-CLASSIFIABLES, 
UNKNOWNS, BLANKS AND OTHER INVALID ENTRIES WILL 
BE ASSIGNED '1s' 
***********************************************/ 
   if not ('02' le gest le '44') then gest1 = 1;  
   else gest1 = 0;  
   if not ('01' le menda le '31') then menda1 = 1;  
   else menda1 = 0; 
   if not ('01' le menmo le '12') then menmo1 = 1;  
   else menmo1 = 0;  
   if menyr in ('9999', ' ', '0000') then menyr1 = 1;  
   else menyr1 = 0; 

   if not ('00' le care le '09') then care1 = 1;  
   else care1 = 0;  
   /*CODE ONLY IF "MONTH PRENATAL CARE BEGAN" IS BLANK*/  
   if care eq '' and not ('01' le nammo le '12') then nammo1 = 1; 
   else nammo1 = 0; 
   if not ('00' le visits le '97') then visits1 = 1;  
   else visits1 = 0; 
   if wtgain_ in (999, .) or wtgain_ gt 200 then wtgain1 = 1;  
   else wtgain1 = 0; 
   if gest1 + menda1 + menmo1 + menyr1 +  
      care1 + nammo1 + visits1 + wtgain1 gt 0 then rec_err = 1; 
   else rec_err = 0; 
RUN; 
 
The final IF-THEN statement creates a new variable (rec_err) which is coded “1” 
if a birth record contains at least one invalid entry in the selected fields. Clean 
birth records are coded “0”.  
      
Step 1: The Summary Procedure 
 
/************************************************ 
STEP 1: THE SUMMARY PROCEDURE SUMMARIZES THE DATA  
BY COMPUTING COUNTS AND PERCENTAGES 
*************************************************/ 
PROC SUMMARY data = temp; 
   where ostate eq '004';   
   class bhosp_;  
   var gest1 -- rec_err;  
   output out = summary  
   (drop = _type_ rename = (_freq_ = Birth)) 
   sum = mean = / autoname; 
RUN; 
 
As we are interested in Arkansas birth hospitals only, the WHERE statement 
requests that the summarization be limited to births that occurred in Arkansas 
and bhosp_ (Birth Hospital Code) is the class variable in the CLASS statement. 
The range of variables listed in the VAR statement represents the selected birth 
certificate items (total = 8) for the report together with the newly created rec_err 
(true if record contains at least one invalid entry) - a total of 9 variables.  
 
Request the output of the SUMMARY procedure be routed to a work dataset 
(summary) by using the OUTPUT OUT = statement. Since the automatically 
generated variable _TYPE_ has no use in this scenario (only one variable in the 
CLASS statement), we used DROP = option to drop it from the outputted dataset 
“summary”. You can use RENAME = option to rename another automatically 

generated variable in PROC SUMMARY, _FREQ_ to “birth”4 to indicate the 
number of births for each level in the CLASS variable. The SUM = and MEAN = 
options are the requested statistics for the variables listed in the VAR statement. 
They compute the counts and percentages, respectively. Note that the word 
“mean” may indicate statistical average but will work in this scenario since the 
data is categorical (1s or 0s). The AUTONAME option assigns the strings _SUM 
and _MEAN to generated variable names containing computed frequencies and 
proportions. We could have used the AUTOLABEL option as well, but the 
flexibility of DEFINE statements in PROC REPORT (discussed later in this 
paper) gives you more control on the appearance of column headers.     
 
Step 2: Round and Sort 
 
/********************************************************** 
STEP 2: THIS DATA STEP AND ITS SUBSEQUENT ARRAY 
STATEMENT AND ROUND FUNCTION ROUND THE COMPUTED 
PERCENTAGES IN STEP 1 INTO 4-DIGITS AFTER THE DECIMAL POINT  
***********************************************************/ 
DATA report1 (drop = i);  
   set summary; 
   array new (9) gest1_mean -- rec_err_mean;  
   do i = 1 to 9;  
   new (i) = round (new (i), 0.0001)*100;  
   end; 
   if bhosp_ eq 'Non-Hosp' then group = 1;  
   else group = 0; *FORCES 'NON-HOSP' TO BE AT THE END ONCE SORTED; 
 
PROC SORT data = report1;  
   by group descending birth; 
RUN; 
 
The ARRAY statement in the previous DATA step utilizes the ROUND function to 
round the computed percentages to 4 digits after the decimal. The multiplication 
by 100 will display the data in a more readable format. Since larger medical 
centers are of more interest to users of such report (they submit larger numbers 
of birth certificates), you want to maintain them at the beginning of the report. 
But, at the same time, you may want to keep the “non-hospital” group at the 
bottom. To do that, create the temporary variable “group” (drop later) in the 
DATA step and then use PROC SORT to sort the data by group first (ascending 
by default-keeps the string “non-hosp” after the numbers in a character variable) 
and then by the total number of births “birth”. Request a descending sort by 
placing the DESCENDING option before “birth”. This keeps hospitals with larger 
numbers of births at the top.  
 

                                                 
4 Or, you can leave the variable _FREQ_ in dataset “summary” in step 1 and label it using DEFINE statement in step 5 

Step 3: Compare to Standard 
   
/************************************************************* 
STEP 3: IN THIS DATA STEP, WE CREATE 3 NEW VARIABLES: 
1. "SCORE85" WHICH REPRESENTS THE NUMBER OF VARIABLES  
    WHERE A HOSPITAL RATE OF REPORTING NON-CLASSIFIABLE, 
    UNKNOWN, OR BLANK DATA FIELDS IS 85% TO EQUAL TO STANDARD, 
2. "SCORE100": # OF VARS WHERE...EXCEEDS NCHS STANDARD AND 
3. "EBC" WHICH REPRESENTS WHETHER THE HOSPITAL USES  
   "ELECTRONIC BIRTH CERTIFICATE" SYSTEM OR NOT 
**************************************************************/ 
DATA report2 (drop = group i);  
   set report1; 
   array scorex (8) gest1_mean -- wtgain1_mean;  
   array standard (8) _temporary_ (1 17.78 5.64 5.03 2.24 2.24 2.99 7.79); 
 
   score85  = 0;  
   score100 = 0;  
 
   do i = 1 to 8;  
      if 0.85*standard (i) le scorex (i) le standard (i)  
      then score85 = score85 + 1;  
      else if scorex (i) gt standard (i)  
      then score100 = score100 + 1;  
   end; 
 
   if bhosp_ in ('HOSPITAL-01', 'HOSPITAL-02', 'HOSPITAL-03', 
                 'HOSPITAL-04', 'HOSPITAL-05', 'HOSPITAL-06', 
                 'HOSPITAL-0X');  
   then EBC = 'YES'; 
   else EBC = 'NO';  
 
   if bhosp_ eq '' then do;  
      bhosp_ = 'Total'; 
      EBC    = 'N/A';  
   end; 
 
   if birth le 5 then delete; 
RUN; 
 
ARRAY “scorex” catalogs the percentages of the 8 selected DATA step variables 
while “standard” records the NCHS standard percentages. Note that we used the 
option _TEMPORARY_ to create the list of standard percentages as temporary 
data elements. You can create new variables that analyses the characteristics of 
hospitals such as rural vs. urban, secondary vs. tertiary, etc. Here, we are using 
the IF-THEN argument again to classify hospitals as “Electronic Birth Certificate” 
vs. “Non-Electronic Birth Certificate” hospitals. Delete hospitals reported less 
than 5 births during the period of evaluation.  

Step 4: Set Ranges for Color-Coding using PROC FORMAT    
 
/********************************************************************** 
STEP 4: THE NEXT FORMAT PROCEDURE ASSIGNS COLORS BASED ON A DECISION RULE: 
*ORANGE IF THE PERCENT OF A SPECIFIC VARIABLE IS 85% - EQUAL TO THE ACCEPTABLE NCHS 
STANDARD AND, 
*RED IF IT EXCEEDS THE NCHS STANDARD 
**********************************************************************; 
PROC FORMAT;  
   value gest   0.85-1      = orange 1.01-high  = red other = black;  
   value menda  15.11-17.78 = orange 17.96-high = red other = black;  
   value menmo  4.79-5.64   = orange 5.70-high  = red other = black;  
   value menyr  4.28-5.03   = orange 5.08-high  = red other = black;  
   value care   1.90-2.24   = orange 2.26-high  = red other = black;  
   value nammo  1.90-2.24   = orange 2.26-high  = red other = black;  
   value visits 2.54-2.99   = orange 3.02-high  = red other = black;  
   value wtgain 6.62-7.79   = orange 7.87-high  = red other = black;  
RUN; 
 
You can compute these ranges in SAS but we found Excel of great utility for such 
computations. The following figure is from the Excel spreadsheet where the 
computations in the FORMAT procedure were obtained.   
  

 
 
For the 85% to equal standard, the lower values for each variables were obtained 
by the Excel formula “=0.85*B3”. The upper values are the standards. For the 
above standard coding, lower values were obtained by the formula “=1.01*B3” 
while the upper values were set to “high”.  

Step 5: Apply the FORMAT and other ODS style Definitions within PROC 
REPORT 
 
 
 
 
 
 
ods listing close;  
ods pdf file = "&path\SCSUG2004.pdf" notoc style = sasweb; 
PROC REPORT data = report2 nowindows                      
   style (header)=[foreground=yellow font_weight=bold font_size=8pt]; 
 
   column bhosp_ birth EBC score100 score85 rec_err_sum rec_err_mean  
          gest1_sum gest1_mean menda1_sum menda1_mean menmo1_sum   menmo1_mean  
          menyr1_sum menyr1_mean care1_sum care1_mean nammo1_sum nammo1_mean  
          visits1_sum visits1_mean wtgain1_sum wtgain1_mean; 
 

(1) Turn off the standard LISTING destination 
and activate a PDF destination with a 
SASWEB STYLE, which by default, 
generates a white FOREGROUND in column 
headers. Include the NOTOC option in the 
ODS PDF destination statement if you do not 
need a table of contents in your PDF file (2) But, PROC REPORT 

gives you more control 
on the color of 
FOREGROUND, weight 
of FONT_WEIGHT, and 
size of FONT_SIZE. 

(3) Like a 
VAR 
statement in 
PROC 
PRINT, 
COLUMN 
statement 
selects the 
arrangement 
of columns 
you want to 
present  

 
   define bhosp_/'Birth/Hospital';  
   define birth/'Birth/Count';  
 
   define score100/'Above/NCHS/Standard' 
      style(header)=[foreground=red];  
 
   define score85/'85% to/Equal/Standard' 
      style(header)=[foreground=orange]; 

(4) DEFINE statements describe how to use and 
display a report item. With the STYLE(item) = 
you can even control style elements (in ODS) for 
report items such as column headers. Here, we 
are requesting the column header for the variable 
“score100” be displayed in red, while column 
header for “score85” be displayed in orange. This 
serves as a “legend” of how to read the color-
coded values in the report.     

 
   define rec_err_sum/'Records/with at/Least/1 Error';   
   define rec_err_mean/'Percent';   
 
   define gest1_sum/'Clinical/Estimate/of/Gestation';  
   define gest1_mean/'NCHS/Standard/=1.00%'  
      style(column)=[foreground=gest. font_weight=bold] 
      style(header)=[font_style=italic]; 
 
   define menda1_sum/'Day/Last/Normal/Menses'; 
   define menda1_mean/'NCHS/Standard/=17.78%'  
      style(column)=[foreground=menda. font_weight=bold] 
      style(header)=[font_style=italic]; 

(5) Remember, we did not use the 
AUTOLABEL option in PROC 
SUMMARY in step 1 because we 
wanted to take advantage of DEFINE 
statements. Here, we are inserting 
NCHS standards in the labels of the 
columns that contain the percentages. 
This works as a reference for 
comparison. Furthermore, we are 
applying the color-coding format (step 
4) using the STYLE (COLUMN)= 
[FOREGROUND=format-name. option  

 
   *CONTINUE DEFINING THE REST OF THE VARIABLES THE SAME WAY; 
 
   title 'NATALITY NOT CLASSIFIABLE, UNKNOWN, AND BLANK DATA BY HOSPITAL: ARKANSAS 2003'; 
RUN;  
ods pdf close;  
ods listing; 
quit; 
 

(6) Add a title to your report using the TITLE statement. Always place 
the RUN statement before ODS statements. Do not forget to turn off 
the ODS PDF destination and turn on the LISTING destination.  

THE FINAL PRODUCT 
The next figures capture the generated report. The report is printed in a legal size 
paper with a landscape orientation. 

CONCLUSION 
In Maternal and Child Health (MCH) research, edited and complete birth data 
produces more accurate and significant results. Arkansas Center for Health 
Statistics and the Division of Vital Records are highly rated from the National 
Center for Health Statistics (NCHS) in regard to the accuracy and completeness 
of their data. To carry out quality MCH research for Arkansas citizens and to 
maintain these high ratings, the center evaluates the performance of birth 
hospitals concerning submitting invalid data in birth certificates. To accomplish 
that, they are putting SAS and its Output Delivery System (ODS) to use.          
 
The introduction of ODS motivated us to move from the multi-step SAS-Excel 
process of generating color-coded data quality reports by hospital of birth to an 
all-in-one process using PROC REPORT. The REPORT procedure combines 
features of the PRINT, MEANS, and TABULATE procedures with features of the 
DATA step in a single report-writing tool. In this project, we are not using all of 
these features, nevertheless; the incorporation of color-coding formats into 
PROC REPORT is of exceptional utility for users of periodical reports in which 
certain levels of performance need to be highlighted.  
 
The default features provided by ODS in the different versions of SAS are 
convenient and can fit any report. To create customized reports, however, you 
may need to alter or modify defaults using the flexibility of ODS style definitions. 
PROC REPORT is an excellent venue to use these style definitions.       
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 39 



REFERENCES 
Karp, Andrew. “ODS 101: Essential Concepts of the SAS® Output Delivery 
System”, Sierra Information Services, Inc. Sonoma, California 
http://www.SierraInformation.com
 
http://www.cdc.gov/nchs/nvss.htm
 
SAS Institute Inc. 2002, SAS 9 Output Delivery System: User’s Guide. Cary, NC 
 
CONTACT INFORMATION 
Your comments and questions are valued and encouraged. Contact us at: 

Phone: (501) 280-4812 
Email: tbaroud@healthyarkansas .com 

Phone: (501) 661-2497 

Paul Johnson 

